$10.00

ISBN 0-944889-08-5



HP42S
EDM Slope Staking

D'Zign

NOTICE

No express or implied warranty is made by D'Zign Land Survey & Development or the author with regard to the procedures and program material offered or their merchantability or their fitness for any particular purpose. The keystroke procedures and program material are made available solely on an "as-is" basis, and the entire risk as to their quality and performance is with the user. Should the procedures or program material prove defective, the user (and not D'Zign Land Survey & Development nor any other party) shall bear any and all cost of all necessary correction and all incidental or consequential damages. D'Zign Land Survey & Development and/or the author shall not be liable for any incidental or consequential damages in connection with or arising out of the furnishing, use, or performance of the keystroke procedures or program material.

HP42S
EDM Slope Staking

Library of Congress Catalog Card No. 89-81214
ISBN 0-944889-08-5

Manufactured in the United States of America
2nd Printing June 1991
3rd Printing August 1992

published by D'Zign Land Survey & Development
Tollhouse, California, U.S.A. 1992

ISBN 0-944889-08-5

TECHNICAL ASSISTANCE

The program material, instructions and procedures contained in this book assume that the user has a working knowledge of both surveying and the general operation of the HP-42S calculator.

Technical assistance is limited to verification of the results shown in the various examples used in the book.

user support

We want you to get maximum performance from your D'Zign product. If you have questions or if you should encounter any difficulties, we will be glad to help! Most of the time you will find the answers right in your Owner's Manual, so check there first. If you still have questions, call for help. Our Support Staff is available at (209) 297-8025 between 8:00 A.M. and 12:00 noon PACIFIC time.

NOTICE

Hewlett-Packard has produced a really powerful calculator at a very good price, the HP-42 Scientific Calculator, which lends itself nicely to solving surveying problems. It can not be programmed by insertion of a module, or with a card reader, like the HP-41 series, but it has a really simple system for typing in a program.

the operations index

To find a function for the first time, HP has provided an "Operations Index" on pages 310 through 335 of the instruction manual, which tells you exactly what keystrokes to use to type in the function you want.

Even better, this index gives you the page number that you can refer to if you want to know more about the function you are using. If, while typing in a program, you aren't sure how to input a particular function, simply refer to the Operations Index.

the softkey menus

All of the programs in this booklet take advantage of the "softkey" menu system built into this calculator. When you want to start a program you stroke **XEQ** and then the softkey corresponding to the program you want, from the menu displayed in the bottom half of the screen.

the programs

There are several different ways to slope stake, and we've tried to make this program as convenient as possible for everyone. You can use it as is, or maybe change some of the output labels to match what is used in your particular area.

The main program is set up to allow setting of a catch point at any convenient location, and the station at which the stakes are set is output. There is also a subroutine which lets you precalculate the angle and distance to a specific station, at any offset, to begin at.

It's a good idea to take your time with the input of the program steps, to avoid errors. You might also want to do the programming in stages, rather than all at one sitting. Be extra careful with the proof-reading.

subroutines

Because of the way the calculator works, we will start by input of some subroutines.

Once the subroutine has been input, its name appears in the menu when you stroke [XEQ], and all you have to do to add it as a step in the program you are typing in is stroke GTO or XEQ followed by keystroking the key corresponding to the subroutine to input the program step GTO XXX or XEQ XXX.

```
00 { 113-Byte Prgm }
01▶LBL "YN"
02 "YES"
03 KEY 1 GTO 01
04 "NO"
05 KEY 2 GTO 02
06 MENU
07 STOP
08▶LBL 01
09 SF 10
10 GTO 03
11▶LBL 02
12 CF 10
13▶LBL 03
14 CLMENU
15 EXITALL
16 RTN
```

getting started

Begin by stroking the shift key, then the **XEQ** key. The display will show a menu which will be blank (if you haven't yet input any programs) except for **.END.** on the left. The keys just below each of the menu portions will correspond to the menu instruction above it. Stroke the key just below the **.END.** in the display.

```
00▶{ 0-Byte Prgm }
01 .END.
```

Next, go into **program mode** by stroking the shifted [R/S] key, and you should have a display similar to the one shown to the left. Begin typing in the program "**YN**".

quick tip

Program steps 03 and 05 use a function which stores the prompt to the menu, and at the same time assigns the key.

To access the function, stroke [ ] [PGM.FCN] [▲] [ ]

You'll receive a prompt, **KEY _**. Stroke the key number (we'll use 1 as the example), and you will get a prompt, **KEY 1 GTO _ _**. Now type in the 01.

next subroutine

This one has three steps we might want to review before you start:

```
09 ARCL ST X
```

To use a store or recall function involving the stack, stroke [STO] (or RCL) [.]. This brings up a menu from which you can select the function to complete the program step.

```
05 1E2
```

Third row down, second from the right, is a key, [E]. Stroke this key, then the number (in this case 2). Nothing will happen until you input the next step, usually 'times' or 'divide'.

```
11 ⊢"+"
```

The ⊢ symbol is "append", which adds to what is already in the **alpha** register.

```
01▶LBL "STA"
02 CF 29
03 FIX 00
04 STO 21
05 1E2
06 ÷
07 ENTER
08 IP
09 ARCL ST X
10 -
11 ⊢"+"
12 FIX 03
13 1E2
14 X
15 10
16 X>Y?
17 ⊢"0"
18 ARCL ST Y
19 RCL 21
20 SF 29
21 FIX 04
22 END
```

This program changes the output of the stationing to really read like stationing. It takes up 48 bytes of memory.

After you've input the program make sure to proof read it. Look, in particular, for steps that are **alpha** but shouldn't be, or should be **alpha** and aren't.

Follow the same procedure to begin the input as with the first subroutine, "**YN**", and type in the steps. When you input the **END** at step 22, do it by stroking XEQ before going into **alpha mode** to type in the rest.

testing

After you do the proof-reading, you can give it quick check, to see if it's working properly. First, clear the alpha register, then input 1204.78 and execute the program. Now, if you go into alpha mode, or execute AVIEW, the display portion should read as 12+04.780.

If you go to the program, "**STA**", which was just input and scroll up to 00, you can begin the next subroutine. When you input the last step, END, the calculator will automatically divide the two programs.

```
01▶LBL "CL1"
02 ΣREG 00
03 CLΣ
04 ΣREG 11
05 CLΣ
06 ΣREG 22
07 CLΣ
08 RTN
09▶LBL "FCL"
10 0.013
11▶LBL "FN"
12 CF IND ST X
13 ISG ST X
14 GTO "FN"
15 END
```

You could also use a RTN, instead of END as step number 15. This leaves the two programs together but, since neither of them uses any subroutine lables, they will both still work properly.

"**CL1**" is a register clearing routine. The other, "**FCL**", clears all of the flags 00 through 13.

This group of programs uses 44 bytes of memory for both programs.

By now you should be familiar with most of the strokes used in the programming. The next subroutine (below) outputs angles labeled with the °, ' and " symbols.

All of these subroutines can be used with other programs. In this one, if flag 19 is set the program clears the alpha register before it starts. When flag 19 is clear alpha labels ahead of the numeric portion are retained.

```
01▶LBL "DMS"
02 FS?C 19
03 CLA
04 ENTER
05 STO 19
06 IP
07 CF 29
08 FIX 00
09 ARCL ST X
10 ⊢"°"
11 -
12 100
13 X
14 ABS
15 STO 18
16 IP
17 XEQ 01
18 ARCL ST X
19 ⊢"'"
20 RCL 18
21 FP
22 100
23 X
24 FIX 02
25 RND
26 FIX 01
27 XEQ 01
28 ARCL ST X
29 ⊢"""
30 CLX
31 FIX 04
32 SF 29
33 RCL 19
34▶LBL 01
35 10
36 X<>Y
37 X<Y?
38 ⊢"0"
39 END
```

Take a few minutes to look the program over after you finish with the input. Like "**STA**", "**DMS**" may be checked by using it. Input 12.34567 and stroke XEQ DMS. If you execute AVIEW, or go into alpha, the display should be 12°34'56.7". If it isn't, something is wrong with one or more of the program steps.

When a program is long it runs slower. The search for a specific label goes all of the way around the loop until it comes to the label it's looking for. For this reason, the input prompting sequences have been set up as a separate program, "**INPT**", which is the next to be put in.

important strokes

The symbol, ↳, is "line feed", and we use it to control the display (the first time we use it is at program step 04). You can input it by stroking

This time, put the program in at the permanent **.END.** and it will have the secondary effect of removing **.END.** from your GTO/XEQ menu. The programs should be grouped in such a way that the ones you use most often appear in the first menu, for convenience. Enter **program mode** and scroll up to 00 to begin.

```
00 { 512-Byte Prgm }
01▶LBL "INPT"
02 CF 21
03 "Instrument"
04 ⊢" Station?↳"
05 PROMPT
06 STO 05
07 "Offset?↳"
08 PROMPT
09 STO 06
10 RCL 05
11 X<0?
12 XEQ 06
13 "Backsight"
14 ⊢" Station?↳"
15 PROMPT
16 STO 03
17 "Offset?↳"
18 PROMPT
19 STO 04
20 RCL 03
21 X<0?
22 XEQ 02
23 RCL 03
24 RCL 04
25 RCL- 06
26 X<>Y
27 RCL- 05
28 →POL
29 CLX
30 X<>Y
31 X<0?
32 360
33 +
34 STO 07
35 "Inst. H.I.?↳"
36 PROMPT
37 STO 11
38 "Profile Elev.?↳"
39 PROMPT
40 STO 20
41 STO 33
42 ENTER
43 ENTER
44 "Cut Factor?↳"
45 PROMPT
46 +
47 STO 27
48 R↓
49 "Fill Factor?↳"
50 PROMPT
51 +
52 STO 28
53 "% Grade?↳"
54 PROMPT
55 1E2
56 ÷
```

```
57 STO 10
58 CF 21
59 "Vertical Curve?"
60 AVIEW
61 SF 21
62 CLMENU
63 XEQ "YN"
64 FC? 10
65 GTO 09
66 SF 04
67 "B.V.C."
68 ⊢" Station?↳"
69 PROMPT
70 STO 16
71 "Length V.C.?↳"
72 PROMPT
73 STO 17
74 "Grade Out?↳"
75 PROMPT
76 1E2
77 ÷
78 STO 09
79 GTO 09
80▶LBL 00
81 CF 01
82 "B.C. Station?↳"
83 PROMPT
84 X<0?
85 SF 01
86 X<0?
87 +/-
88▶LBL 07
89 CLMENU
90 STO 15
91 FS? 01
92 RTN
93 "Radius?↳"
94 PROMPT
95 STO 14
96 "Delta?↳"
97 PROMPT
98 →HR
99 X<0?
100 SF 05
101 STO 13
102 →RAD
103 RCLX 14
104 ABS
105 RCL+ 15
106 STO 24
107 RCL 13
108 ABS
109 2
110 ÷
111 TAN
112 RCLX 14
113 STO 26
114 -
115 STO 25
116 RCL 15
117 STO+ 26
118 RTN
119▶LBL 01
120 R↓
121 CLX
122 RCL 14
123 PI
124 X
125 RCL 15
126 RCL 05
127 XEQ 03
128 RCL 06
129 XEQ 04
130 STO 06
131 X<>Y
132 RCL+ 15
133 STO 05
134 RTN
135▶LBL 02
136 +/-
137 STO 03
138 XEQ 00
139 XEQ 05
140 RTN
141▶LBL 03
142 -
143 +/-
144 180
145 X
146 X<>Y
147 ÷
148 RTN
149▶LBL 04
150 FC? 05
151 +/-
152 RCL+ 14
153 →REC
154 +/-
155 RCL+ 14
156 FS? 05
157 +/-
158 RTN
159▶LBL 05
160 RCL 14
161 PI
162 X
163 RCL 15
164 RCL 03
165 XEQ 03
166 RCL 04
167 XEQ 04
168 STO 04
169 X<>Y
170 RCL+ 15
171 STO 03
172 RTN
173▶LBL 06
174 +/-
175 STO 05
176 XEQ 00
177 XEQ 01
178 RTN
179▶LBL 09
180 FC? 01
181 RTN
182 CLMENU
183 CF 01
184 CF 21
185 "NEXT"
186 AVIEW
187 SF 21
188 "B.C."
189 KEY 1 GTO 07
190 "END@"
191 KEY 6 GTO 08
192 MENU
193 STOP
194▶LBL 08
195 +/-
196 SF 01
197 GTO 07
198 .END.
```

Proof read the program at least once, looking for typos. One quick check is to execute the program and run/stop through the prompts. This will check the **sequence** of the prompts, and should run as follows:

First, execute "**INPT**", to bring up the first prompt.

stroke R/S after each of the prompts appears, until you get to the beginning of the prompts for the vertical curve.

Answer this one "yes" to bring up the remaining vertical prompts, and finally, the prompt for "**NEXT**" and the prompt bar for selection of either an ending point or a B.C. station.

When you stroke R/S at this prompt the display should clear. Go into program mode, and your program pointer should be at step 93, "Radius?↳".

Exit and stroke R/S twice to bring up the next prompt, and continue. A R/S after "Delta?" will not bring up another prompt. If you go into program now, you should be at step 120, LBL 01.

```
Instrument Station?  R/S
Offset?              R/S
Backsight Station?   R/S
Offset?              R/S
Inst. H.I.?          R/S
Profile Elev.?       R/S
Cut Factor?          R/S
Fill Factor?         R/S
% Grade?             R/S
Vertical Curve?      R/S
YES NO                YES
B.V.C. Station?      R/S
Length V.C.?         R/S
Grade Out?           R/S
NEXT
B.C.          ENDe   R/S
☐ R/S  93 "Radius?↳"
                EXIT R/S R/S
Radius?              R/S
Delta?               R/S
☐ R/S  120▶LBL 01    EXIT
```

the main program, "SS"

The best place to put this next program is at the bottom of "**INPT**". GTO "**INPT**" and enter **program mode**. Scroll upward 3 times, to put the pointer at step 197, GTO 07, and type in the first step of the new program, **LBL "SS"**.

Next, scroll upward to GTO 07 again, stroke XEQ and then enter **alpha mode** and type in "END". This separates the programs and positions the new one correctly. Leave **program mode** and GTO "**SS**" to input the rest of the program listing which begins on page 8.

Take your time with the programming, because this is a rather long one, and extremely difficult to debug.

```
00 { 1008-Byte Prgm }
01▶LBL "SS"
02 XEQ "CL1"
03 XEQ "FCL"
04 CLST
05 SF 01
06 XEQ "INPT"
07▶LBL 00
08 CLX
09 CLMENU
10 "H<"
11 KEY 1 GTO 01
12 "Z<"
13 KEY 2 GTO 01
14 "S.D."
15 KEY 3 GTO 01
16 "C"
17 KEY 4 GTO C
18 "F"
19 KEY 5 GTO F
20 "ROD"
21 KEY 6 GTO A
22 KEY 7 GTO 02
23 KEY 8 GTO 02
24 KEY 9 GTO 03
25 MENU
26▶LBL 04
27 STOP
28 GTO 04
29▶LBL 02
30 CLMENU
31 "STA"
32 KEY 1 GTO 06
33 "O/S"
34 KEY 2 GTO 07
35 "-USE-"
36 KEY 3 GTO 12
37 "NEWC"
38 KEY 4 GTO 14
39 "NEWF"
40 KEY 5 GTO 25
41 "REF"
42 KEY 6 GTO 09
43 KEY 7 GTO 00
44 KEY 8 GTO 00
45 MENU
46▶LBL 05
47 STOP
48 GTO 05
49▶LBL 03
50 CLMENU
51 EXITALL
52 STOP
53 RTN
54▶LBL 01
55 CF 22
56 STOP
57▶LBL 06
58 STO 30
59 CLA
60 CF 21
61 XEQ "STA"
62 AVIEW
63 SF 21
64 STOP
65▶LBL 14
66 RCL+ 33
67 STO 27
68 STO 20
69 STOP
70▶LBL 25
71 RCL+ 33
72 STO 28
73 STO 20
74 STOP
75▶LBL 17
76 FIX 02
77 CLA
78 "RP Elev = "
79 ARCL ST X
80 ⊢"↳"
81 RCL- 12
82 +/-
83 X>0?
84 ⊢"+"
85 ARCL ST X
86 ⊢"@ "
87 ARCL 34
88 ⊢" to S.S."
89 AVIEW
90 CF 06
91 ADV
92 GTO 00
93▶LBL A
94 SF 03
95 FC? 22
96 RCL 01
97 FS? 22
98 +/-
99 STO 01
100 R↓
101 X<>Y
102 →HR
103 X<>Y
104 →REC
105 RCL+ 11
106 RCL+ 01
107 FS? 06
108 XEQ 17
109 FS?C 06
110 GTO 10
111 STO 12
112 R↓
113 X<>Y
114 →HR
115 RCL+ 07
116 X<>Y
117 →REC
118 RCL+ 05
119 STO 30
120 X<>Y
121 RCL+ 06
122 STO 31
123 X<>Y
124 RCL 24
125 X≤Y?
126 GTO 08
127 CLX
128 RCL 15
129 FS? 01
130 GTO 11
131 X>Y?
132 GTO 11
133 -
134 X<>Y
135 FS? 05
136 +/-
137 RCL- 14
138 +/-
139 →POL
140 +/-
141 RCL+ 14
142 FS? 05
143 +/-
144 STO 31
145 X<>Y
146 →RAD
147 RCLX 14
148 RCL+ 15
149 STO 30
150 RCL 24
151 X≤Y?
152 GTO 08
153 R↓
154 X<>Y
155 FS? 05
156 +/-
157 X<>Y
158 ENTER
159 GTO 11
160 RTN
161▶LBL 27
162 RCL 08
163 "W/2 = "
164 ARCL ST X
165 ⊢"↳"
166 PROMPT
167 STO 08
168 SF 08
169 RCL 29
170 ENTER
171 IP
172 X=Y?
173 CF 08
174 R↓
175 FIX 00
176 CF 29
177 FS? 08
178 FIX 02
179 "Slope Ratio = "
180 ARCL ST X
181 ⊢":1↳"
182 SF 21
183 PROMPT
184 FIX 02
185 SF 29
186 STO 29
187 RCL 23
188 RCL- 12
189 STO 02
190 ABS
191 RCLX 29
192 RCL+ 08
193 STO 32
194 RCL 31
195 X<0?
196 +/-
197 -
198 +/-
199 STO 00
200 CLA
201 "TRY "
202 ABS
203 ARCL ST X
204 RCL 31
205 X>0?
206 XEQ 29
207 RCL 00
208 X<0?
209 ⊢"' LEFT"
210 X>0?
211 ⊢"' RIGHT"
212 CF 21
213 AVIEW
214 SF 21
215 XEQ 02
216 RTN
217▶LBL 29
218 RCL 02
219 X>0?
220 RTN
221 -1
222 STOX 00
223 RTN
224▶LBL 12
225 FIX 01
226 CLA
227 "FILL "
228 RCL 02
229 X<0?
230 "CUT "
231 ABS
232 ARCL ST X
233 ⊢"↳"
234 RCLX 29
235 ⊢" AT "
236 ARCL ST X
237 AVIEW
238 CF 03
239 RCL 32
240 RCL 30
241 ENTER
242▶LBL 11
243 R↓
244 CLA
245 "Sta "
246 XEQ "STA"
247 RCL 31
248 FIX 02
249 X>0?
250 GTO 13
251 +/-
252 ⊢"↳at "
253 ARCL ST X
254 ⊢" Left"
255▶LBL 15
256 FC? 03
257 AVIEW
258 "Elev. = "
259 ARCL 12
260 ⊢"↳"
261 XEQ 28
262 FS? 03
263 GTO 27
264▶LBL 10
265 CF 21
266 "INPUT SHOT"
267 AVIEW
268 SF 21
269 GTO 00
270▶LBL 13
271 ⊢"↳at "
272 ARCL ST X
273 ⊢" Right"
274 GTO 15
275▶LBL 28
276 RCL 30
277 FS? 04
278 GTO 21
279▶LBL 26
280 RCL 30
281 RCL- 05
282 RCLX 10
283 RCL 20
284▶LBL 24
285 +
286 STO 23
287 ⊢"Grade = "
288 ARCL ST X
289 FC? 03
290 AVIEW
291 ADV
292 FS? 03
293 GTO 27
294 GTO 02
295▶LBL 09
296 SF 06
297 ENTER
298 ABS
299 STO 34
300 R↓
301 RCL+ 31
302▶LBL 07
303 RCL- 06
304 RCL 30
305 RCL 24
306 X≤Y?
307 GTO 20
308 CLX
309 RCL 15
310 X≤Y?
311 XEQ 19
312 R↓
313 RCL- 05
314▶LBL 16
315 →POL
316▶LBL 18
317 FIX 02
318 "DIST = "
319 ARCL ST X
320 ⊢"↳"
321 FIX 04
322 CLX
323 X<>Y
324 RCL- 07
```

```
325 X≤Y?
326 360
327 +
328 →HMS
329 ⊢"∠ = "
330 XEQ "DMS"
331 AVIEW
332 ADV
333 GTO 10
334▶LBL 19
335 -
336 RCL÷ 14
337 →DEG
338 STO 03
339 COS
340 X<>Y
341 FC? 05
342 +/-
343 RCL+ 14
344 X
345 STO 04
346 RCL 03
347 TAN
348 X
349 RCL 15
350 RCL- 05
351 +
352 RCL 04
353 RCL- 14
354 FC? 05
355 +/-
356 X<>Y
357 GTO 16
358▶LBL 20
359 R↓
360 X<>Y
361 RCL+ 06
362 X<>Y
363 RCL- 25
364 →POL
365 X<>Y
366 RCL+ 13
367 X<>Y
368 →REC
369 X<>Y
370 RCL- 06
371 X<>Y
372 RCL 26
373 RCL- 05
374 +
375 →POL
376 GTO 18
377▶LBL 22
378 RCL- 17
379 X>Y?
380 GTO 26
381 GTO 23
382▶LBL 21
383 RCL 16
384 RCL+ 17
385 X>Y?
386 XEQ 22
387 X<>Y
388 RCL 16
389 RCL 17
390 2
391 ÷
392 +
393 ENTER
394 RCL- 05
395 RCLX 10
396 RCL+ 20
397 RCL 30
398 RCL- 16
399 RCL 17
400 2
401 ÷
402 -
403 RCLX 09
404 GTO 24
405▶LBL 23
406 -
407 ENTER
408 ENTER
409 RCL 09
410 RCL- 10
411 1E2
412 X
413 RCL÷ 17
414 X
415 2
416 ÷
417 RCL 10
418 1E2
419 X
420 +
421 X
422 1E2
423 ÷
424 RCL 16
425 RCL- 05
426 RCLX 10
427 RCL+ 20
428 GTO 24
429▶LBL C
430 RCL 27
431 STO 20
432 GTO 10
433▶LBL F
434 RCL 28
435 STO 20
436 GTO 10
437▶LBL 08
438 RCL 31
439 RCL 30
440 RCL- 26
441 →POL
442 X<>Y
443 RCL- 13
444 +/-
445 X<>Y
446 →REC
447 RCL+ 25
448 STO 30
449 X<>Y
450 +/-
451 STO 31
452 X<>Y
453 ENTER
454 GTO 11
455 .END.
```


Before starting with the user's instructions and the keystroke examples, this is a good place to look at how the program is set up.

vertical grade



When the station that is occupied by the instrument is within a vertical curve, the elevation that is input as profile grade is the elevation of the **vertical tangent** at the instrument station.

It's a simple matter to calculate this elevation since it is just the extention of the back vertical tangent (even if the instrument is upstation of the PVI).

slope ratio

The slope ratio is carried as a constant, and is displayed each shot. If you want to change it, input the new number before stroking R/S. If it is the ratio you plan to use just stroke R/S.

half-width

Like the slope ratio, the half-width (distance from the centerline to the hinge point) is displayed each shot. If it is the width you want to use, stroke R/S. If not, input the new number and then stroke R/S.

Both of the variables above may be changed at any time, during the shot input, allowing for widening the roadbed or flattening the slope as the cut or fill approaches a "daylight" area.

referencing

A routine has also been included which calculates the angle to turn and the distance to measure to set the

reference stake. The amount of offset is input by the user and the angle and distance are calculated. A shot taken on the reference hub after it is set will output the elevation of the RP and the plus or minus to the slope stake.

setting specific stations

This program allows the flexibility of setting stakes wherever the terrain forms a high or low point, to best define the top or toe of slope. Since the program calculates the finished grade at any station you shoot, it isn't necessary to only set those points with elevations shown on the profile.

To stake at specific stations, at 25' or 50' intervals, you can set the station at an offset (determined as approximate catch point based on the drawings) for the first trial station, using [STO] and [IND].

After input of the station and offset the program will output the angle and distance to that point. Set a temporary point there and take a shot on it. The output will tell you which way, and how far, to the catch point.

Since the ground changes, it should be understood that this distance is where the catch point would be if the ground were flat.

the cut and fill factors

The actual point where the slope will end is seldom at the same elevation as profile grade. Use of these factors causes the adjustments to profile grade to be made automatically. These factors are discussed in more detail as part of the keystroke instructions.

sizing

To use the basic program, the calculator should be sized to 0035. This means that registers 00 through 34 are used for storage. If you decide to add either or both of the optional routines (resection and data storage) additional registers will have to be allocated for their use.

To begin, the program is initialized by keystroking [ ] [XEQ] [ ]. This clears the registers, resets the status of the flags, and calls up the first prompt:

Instrument Station?

Input the station which (or opposite which) is occupied by the instrument. If this station is within a curve on the alignment, stroke [+/-] before stroking [R/S]

Offset?

If the instrument is on centerline, input 0. If the instrument is at an offset distance from centerline, input the offset distance. If the offset is left of centerline, stroke [+/-] before [R/S]

Backsight Station?

Input the station at the backsight. If the station is within a curve (and the instrument station was not), first stroke [+/-], then [R/S]

Offset?

Input 0 if the sight is on centerline, or input the offset distance. If the offset of the point is to the left of centerline, stroke [+/-] before [R/S]

B.C. Station?

This prompt will appear after the instrument station input, if the instrument is on a curve in the alignment, or after input of the backsight information, if it is on a curve. If there is a curve, the prompts marked * will appear.

If neither point was on a curve, but there is a curve in the work area, a later prompt provides the opportunity to input the curve data. Input the station, [R/S]

Radius*

Input the length of the radius of the curve

R/S

Delta?*

Input the central angle of the curve. If the curve is a curve to the left, +/- before stroking

R/S

Inst. H.I.?

Input the elevation of the height of instrument. Stroke

R/S

Profile Elev.?

Input the **finished grade** elevation at the instrument station. If the instrument is at a station which is located within a vertical curve, input the elevation of the **back tangent profile grade.**

R/S

Cut Factor?

Input the "cut factor". This is the difference in elevation between the finished (profile) grade and the hinge-point of the cut section. If negative, +/- , then stroke

R/S

Fill Factor?

Input the "fill factor". This is the difference between profile grade and the hinge-point in a fill section. If negative, +/- and stroke

R/S

% Grade?

Input the percent of grade for the vertical alignment. If negative, +/-. Stroke

R/S

Vertical Curve?

YES NO

This prompt requires a yes or no answer. If the answer is "YES", the next three prompts for input of the vertical data will appear:

B.V.C. Station?

Input the station at the beginning of the vertical curve. In the case of a grade break, instead of a curve, input the station at the grade break

R/S

Length V.C.?

Input the length of the vertical curve. If this is a grade break instead of a curve, input 0

R/S

Grade Out?

Input the percent of grade leaving the vertical curve. If negative, +/- before stroking

R/S

NEXT

B.C. ENDC

If there is a curve in the work area, input the B.C. station and stroke B.C.. If there is not, input a station that is as far as you intend to go, to set the auto-stop option, stroke ENDC.

INPUT SHOT

HI ZE S.D. C F ROD

The first time this prompt appears, or at any time when you want to change the hinge point elevation (from cut to fill or from fill to cut) select the appropriate factor

If staking a cut, stroke

If staking a fill, stroke

INPUT SHOT

HI ZE S.D. C F ROD

Input the horizontal angle

Input the zenith angle

Input the measured slope distance

Because, in slope staking, the rod is always a minus the program will indicate that; you need not change the sign

Once the rod has been input you do not have to input it again, unless it changes. Stroking ROD will cause the program to repeat the last rod reading for this shot.

Input the rod reading

The next two prompts are for the "half-width" and the slope ratio. Since they both vary, the value about to be used for the calculation is displayed, and may be changed at this time by input of a new value, prior to stroking R/S.

W/2 = 0.00

Input the correct half-width value, if it is different than the value which is displayed

R/S

Slope Ratio = 0:1

Input the correct slope ratio value, if different from the value displayed

R/S

At this point the display will show the distance to the actual catch point. If the distance is considered to be within tolerance, stroke ▯▯▯. If it is not, have the rodman move to a new position (based on the display), scroll once, with ▲ or ▼ and begin with input of the data from the new shot.

to set a reference stake

Once you have reached the catch point, stroking ▯▯▯ will display the information for making out the slope stake.

Input the distance you want to use, from the slope stake to the reference stake. If it is left of centerline, +/- , then stroke

Output will be the distance to the reference point and the horizontal angle to turn to set it.

INPUT SHOT

After setting the reference point, take a shot on it and input it in the same manner as the previous shots.

Output will be the elevation of the reference point and the difference in elevation to the slope stake.

changing the cut or fill factors

At times, design conditions change and a new "Typical Section" means using a different cut or fill factor. There are keys for doing this in the second menu.

To change the cut factor, input the new factor and stroke

To change the fill factor, input the new factor and stroke

This will calculate the elevations based on the new input until you change it again. The new factor also becomes the **current** factor. If you have just input a new fill factor, but are currently staking a cut stroke ▯ before beginning input of the shot.



Before you begin input of the alignment information, take a moment to calculate the cut and fill factors. This is simply the difference in elevation between the finish profile grade and the hinge point you are staking to. If for instance, you were staking to the back of a bench in a cut section, this could be a difference of 20 or more feet.

We are going to use the roadway details shown on the opposite page for the keystroke examples, so we won't have that type of condition. In the typical cut section we are going to be staking to the flowline of the ditch.

The **factor** will be the sum of the differences. We have 17.5 feet at -2%, +0.5' for the curb, and 8.5' at +4% on both sides; (17.5)(-.02) + 0.5 + (8.5)(.04) = +0.49. This is the **fill factor.** For a cut, we have the added ditch, 4' @ 2:1, which is -2', so the **cut factor** will be -1.51'.

Let's start with the keystrokes for input of the basic information about our setup and the alignments of our roadway. The calculator should be sized at **030**, and we initialize the program by stroking [ ] XEQ [SST].

Instrument Station?

keystrokes:

1 0 0 0 R/S

Offset?

keystrokes:

7 +/- R/S

Backsight Station?

keystrokes:

1 3 7 5 +/- R/S

(it's on a curve)

Offset?

keystrokes:

0 R/S

B.C. Station?

keystrokes

1 2 5 0 R/S

Radius?

keystrokes:

8 0 0 R/S

Delta?

keystrokes:

8 . 5 7 0 9 R/S

Inst. H.I.?

keystrokes:

1 1 3 . 6 2 R/S

Profile Elev?

keystrokes:

1 0 6 . 7 5 R/S

Cut Factor?

keystrokes:

1 . 5 1 +/- R/S

Fill Factor?

keystrokes:

. 4 9 R/S

% Grade?

keystrokes:

1 . 5 +/- R/S

Vertical Curve?

YES NO

keystroke:

YES

B.V.C. Station?

keystrokes:

1 0 5 0 R/S

Length V.C.?

keystrokes:

2 0 0 R/S

Grade Out?

keystrokes:

4 R/S

x:0.00

HI Z SD C F ROD

Prior to the first shot input, we need to say whether it's a cut or fill section. We'll assume that the first shot will be fill, and stroke F.

INPUT SHOT

HI Z SD C F ROD

Let's try a shot using the following data:

horiz. angle = 13°10'20"
zenith angle = 92°05'15"
slope distance = 187.40'
the rod reading is 5.00'

keystrokes:

1 3 . 1 0 2

HI

9 2 . 0 5 1 5

Z

1 8 7 . 4

SD

5 ROD

W/2 = 0.00

keystrokes:

2 6 R/S

Slope Ratio = 0:1

keystrokes:

3 R/S

display: TRY 2.55' RIGHT

You need to have the rodman go about 2½ feet to the right. We take a shot there, with the following data:

horiz. angle = 13°57'00"
zenith angle = 92°42'00"
slope distance = 187.90'
the rod reading is 5.00'

Input this data the same as we did the first shot. This time the prompts for the half-width and the slope ratio will already show us the right information, so we can just continue at that point.

keystrokes:

▲ 1 3 . 5 7

HI

9 2 . 4 2

Z

1 8 7 . 9

SD

ROD

W/2 = 26.00

keystroke:

R/S

Slope Ratio = 3:1

keystroke:

R/S

display: TRY 0.94' LEFT

The ground slope has made a difference, and you need to go left about a foot, at the same elevation, for a catch point. On this shot we get:

horiz. angle = 13°30'25"
zenith angle = 92°27'00"
slope distance = 187.70'
the rod reading is 5.00'

keystrokes:

▼

1 3 . 3 0 2 5

HI

9 2 . 2 7

Z

1 8 7 . 7

SD

ROD

W/2 = 26.00'

keystroke:

R/S

Slope Ratio = 3:1

keystroke:

R/S

display: TRY 0.09' RIGHT

This is close enough to use for the catch point. Stroke

DIRECT

output:

```
FILL 6.3
 AT 18.8
Sta 11+80.203
at 44.90 Right
Elev. = 100.60
Grade = 106.87
```

Let's set the reference point at 15'

R/S

keystrokes:

1 5 REF

output:

```
DIST = 192.22
∠ = 17°48'26.7"
```

INPUT SHOT

HI | ZE | S.O. | C | F | ROD

Turn the angle shown and set the reference point at the distance given. After the point is set, take a shot on it to get the elevation. We'll use the following data:

horiz. angle = 17°48'30"
zenith angle = 93°06'15"
slope distance = 192.20'
the rod reading is 5.00'

keystrokes:

1 7 · 4 8 3 HI

9 3 · 0 6 1 5 ZE

1 9 2 · 2 S.O.

ROD

output:

```
RP Elev = 98.21
+2.38 @ 15.00 to S.S.
```

x:**0.00**

HI | ZE | S.O. | C | F | ROD

Let's try one on the cut side. First, change the factor by stroking C

INPUT SHOT

HI | ZE | S.O. | C | F | ROD

We'll use the following data:

horiz. angle = 348°23'00"
zenith angle = 89°19'50"
slope distance = 201.90'
the rod reading is 5.00'

keystrokes:

3 4 8 · 2 3 HI

8 9 · 1 9 5 ZE

2 0 1 · 9 S.O.

ROD

W/2 = 26.00

keystrokes:

3 0 R/S

Slope Ratio = 3.00

keystrokes:

1 · 5 R/S

display: TRY 0.29' RIGHT

Let's go a little to the right and try another shot. We get:

horiz. angle = 348°25'40"
zenith angle = 89°17'20"
slope distance = 202.30'
the rod reading is 5.00'

keystroke:

x:**0.00**

HI | ZE | S.O. | C | F | ROD

keystrokes: ▲

3 4 8 · 2 5 4 HI

8 9 · 1 7 2 ZE

2 0 2 · 3 S.O.

ROD

W/2 = 30.00

keystroke: R/S

Slope Ratio = 1.50:0

keystroke: R/S

display: TRY 0.02' LEFT

keystroke: DIRECT

output:

```
CUT 5.8
 AT 8.7
Sta 11+99.787
at 38.69 Left
Elev. = 111.13
Grade = 105.33
```

R/S

keystrokes:

1 0 +/- RE

(we're going left)

output:

```
DIST = 204.09
∠ = 345°39'14.3"
```

INPUT SHOT

HI | ZE | S.O. | C | F | ROD

After setting the RP, we get a shot of:

horiz. angle = 345°39'00"
zenith angle = 87°32'00"
slope distance = 204.10'
the rod reading is 5.00'

keystrokes:

3 4 5 · 3 9 HI

8 7 · 3 2 ZE

2 0 4 · 1 S.O.

ROD

output:

```
RP Elev = 117.40
-6.27 @ 10.00 to S.S.
```

x:**0.00**

The shots listed below will give you some additional practice with the keystrokes. The shots are to the catch points, still using the same alignment. Remember to change the cut or fill factor (if necessary) before input of the shot. Forgetting to do so will result in an error at the catch point.

1. **Cut section**, set the RP @ 15'. Use a zenith angle of 88°12'20" for the RP shot.

horiz. angle = 354°40'50"
zenith angle = 88°29'25"
slope distance = 400.70'
the rod reading is 5.00'

```
CUT 6.4
 AT 9.6
Sta 13+94.686
at 39.57 Left
Elev. = 119.18
Grade = 112.78

DIST = 403.87
∠ = 352°35'47.1"

RP Elev = 121.27
-2.09 @ 15.00 to S.S.
```

2. **Fill section,** set the RP @ 15'. Use a zenith angle of 89°40'30" for the RP shot.

horiz. angle = 6°13'55"
zenith angle = 89°25'40"
slope distance = 477.50'
the rod reading is 5.00'

```
FILL 4.8
 AT 14.4
Sta 14+79.939
at 40.40 Right
Elev. = 113.39
Grade = 118.19

DIST = 477.67
∠ = 8°01'53.3"0

RP Elev = 111.33
+2.06 @ 15.00 to S.S.
```

3. **Cut section,** set the RP @ 15'. Use a zenith angle of 87°52'10" for the RP shot.

horiz. angle = 351°23'30"
zenith angle = 88°21'00"
slope distance = 315.00'
the rod reading is 5.00'

```
CUT 8.3
 AT 12.5
Sta 13+09.986
at 42.54 Left
Elev. = 117.69
Grade = 109.39

DIST = 317.39
∠ = 348°42'48.1"

RP Elev = 120.42
-2.73 @ 15.00 to S.S.
```

The two most likely sources of error when using this program are forgetting to change the factor when going from cut to fill (or fill to cut) and not checking that the half-width and slope ratio are correct for the shot being taken.

If you're using the calculator without a printer, remember to continue stroking R/S after output, until the next answer or prompt appears.

resection

This program has been added as an option, for those days when you could stake the whole job if you had a point "up on that hill". It works with the stations and offsets, so you can determine the station and offset of a random point anywhere that you can see three known points from.

The progrom locates the station and offset of the instrument setup point, and then outputs the distances to the points which were used as backsights, as a check on the shots used. All that is required is the input of the Station/Offset information for three points and the angles between the first and second and second and third points.

The best place to put this one is on top of "**SS**". Stroke ▢ XEQ ▒▒ to go to "**SS**", then scroll **up** above the **.END.** to begin. When you have finished the input, the "END" on this program will separate them.

```
00 { 587-Byte Prgm }
01▶LBL "R'SEC"
02 XEQ "CL1"
03 XEQ "FCL"
04 180
05 STO 15
06 SF 07
07 "Sta 1 "
08 ASTO 30
09 ⊢"↑ O/S↳"
10 PROMPT
11 XEQ 03
12▶LBL 07
13 STO 01
14 X<>Y
15 STO 00
16 SF 08
17 "Sta 2 "
18 ASTO 31
19 ⊢"↑ O/S↳"
20 PROMPT
21 XEQ 03
22▶LBL 08
23 STO 03
24 X<>Y
25 STO 02
26 SF 09
27 "Sta 3 "
28 ASTO 32
29 ⊢"↑ O/S↳"
30 PROMPT
31 XEQ 03
32▶LBL 09
33 STO 05
34 X<>Y
35 STO 04
36 "∠ 1st - 2nd?↳"
37 PROMPT
38 →HR
39 STO 11
40 "∠ 2nd - 3rd?↳"
41 PROMPT
42 →HR
43 STO 12
44 RCL 01
45 RCL- 03
46 RCL 00
47 RCL- 02
48 →POL
49 STO 06
50 X<>Y
51 STO 14
52 RCL 05
53 RCL- 03
54 RCL 04
55 RCL- 02
56 →POL
57 STO 07
58 X<>Y
59 RCL- 14
60 RCL- 11
61 RCL- 12
62 STO 10
63 2
64 ÷
65 ENTER
66 TAN
67 RCL 07
68 RCL÷ 06
69 RCL 11
70 SIN
71 X
72 RCL 12
73 SIN
74 ÷
75 1
76 -
77 X
78 LASTX
79 2
80 +
81 →POL
82 R↓
83 +
84 ENTER
85 X<0?
86 XEQ 05
87 STO 08
88 CF 07
89 X<>Y
90 +/-
91 RCL+ 10
92 X<0?
93 XEQ 05
94 STO 10
95 RCL 08
96 RCL 14
97 RCL 11
98 RCL- 15
99 +
100 +
```

```
101 RCL 08
102 SIN
103 RCL× 06
104 RCL 11
105 SIN
106 ÷
107 →REC
108 RCL+ 02
109 STO 16
110 X<>Y
111 RCL+ 03
112 STO 17
113 ADV
114 GTO 14
115▶LBL 06
116 0
117 STO 24
118 30
119 STO 20
120▶LBL 02
121 RCL IND 24
122 RCL- 16
123 ISG 24
124 STO ST X
125 RCL IND 24
126 RCL- 17
127 X<>Y
128 →POL
129 FIX 02
130 "Dist to "
131 ARCL IND 20
132 ⊢" "
133 ARCL ST X
134 AVIEW
135 ISG 20
136 STO ST X
137 ISG 24
138 STO ST X
139 33
140 RCL 20
141 X=Y?
142 STOP
143 ADV
144 GTO 02
145 RTN
146▶LBL 03
147 STO 33
148 X<>Y
149 STO 34
150 X<>Y
151 CF 21
152 "Curve?"
153 AVIEW
154 SF 21
155 XEQ "YN"
156 FS? 10
157 XEQ 10
158 GTO 13
159 RTN
160▶LBL 13
161 FS? 09
162 GTO 09
163 FS? 08
164 GTO 08
165 FS? 07
166 GTO 07
167 RTN
168▶LBL 14
169 RCL 39
170 RCL 16
171 X>Y?
172 XEQ 16
173 RCL 36
174 X<Y?
175 XEQ 15
176 CLA
177 RCL 16
178 "Inst. Located a"
179 ⊢"t Sta."
180 XEQ "STA"
181 RCL 17
182 X<0?
183 SF 01
184 ABS
185 FIX 02
186 ⊢" @ "
187 ARCL ST X
188 FS? 01
189 ⊢" Lt"
190 FC? 01
191 ⊢" Rt"
192 AVIEW
193 CF 08
194 ADV
195 GTO 06
196▶LBL 15
197 -
198 RCL÷ 37
199 →DEG
200 RCL 37
201 →REC
202 RCL- 37
203 FC? 05
204 +/-
205 STO+ 17
206 X<>Y
207 FC? 05
208 +/-
209 STO+ 16
210 RTN
211▶LBL 05
212 RCL+ 15
213 RTN
214▶LBL 10
215 R↓
216 FS? 00
217 XEQ 11
218 FS? 00
219 RTN
220 "B.C.Station?↳"
221 PROMPT
222 STO 36
223 "Radius?↳"
224 PROMPT
225 STO 37
226 "Delta?↳"
227 PROMPT
228 →HR
229 X<0?
230 SF 05
231 STO 38
232 ABS
233 →RAD
234 x
235 RCL+ 36
236 STO 39
237 XEQ 11
238 RTN
239▶LBL 11
240 RCL 39
241 RCL 34
242 X>Y?
243 GTO 12
244 RCL- 36
245 RCL÷ 37
246 →DEG
247 RCL 37
248 →REC
249 RCL- 37
250 FC? 05
251 +/-
252 STO+ 33
253 X<>Y
254 RCL+ 36
255 STO 34
256 RCL 33
257 RTN
258▶LBL 16
259▶LBL 12
260 "PASSED E.C."
261 AVIEW
262 END
```




The example for using the program will use the alignment that is shown to the left. The program is fully prompted, so that the keystrokes example can also serve as the instructions.

Remember to size the calculator to at least size **0040** before beginning, or you'll get a "size error" the first time you try to use a register that isn't there.

Execute **"R'SEC"** to bring up the first prompt.

prompt: Sta 1 ↑ O/S

keystrokes:

1 0 0 0 ENTER

0 R/S

prompt: Curve?

YES NO

keystrokes:

NO

prompt: Sta 2 ↑ O/S

keystrokes:

1 2 0 0 ENTER

0 R/S

prompt: Curve?

YES NO

keystrokes:

NO

prompt: Sta 3 ↑ O/S

keystrokes:

1 3 0 0 ENTER

0 R/S

prompt: Curve?

YES NO

keystrokes:

YES

prompt: B.C.Station?

keystrokes:

1 2 0 0 R/S

prompt: Radius?

keystrokes:

5 0 0 R/S

prompt: Delta?

keystrokes:

2 5 R/S

prompt: ∠ 1st - 2nd?

keystrokes:

2 2 . 2 4 3 R/S

prompt: ∠ 2nd - 3rd?

keystrokes:

1 0 . 5 9 0 7 R/S

output:

```
Inst. Located at Sta.
11+25.543 @ 503.60 Rt

Dist to Sta 1   519.01

Dist to Sta 2   509.07

Dist to Sta 3   523.33
```

Do the usual proof reading after you are done with the program input. Try the example to check the program.

coordinate resection

This same program will do coordinate resection, if you input the north-coordinate as the station and the east-coordinate as the offset.

Answer all of the CURVE? prompts **NO**, and the answer will be the coordinates of the new point, even though they are labled as "station" and offset".

To have more than the two place accuracy, you can fix the display at 04 or 06 places, and then recall the newly computed coordinates directly out of register 16 (north) and 17 (east), where they are stored.




```
Inst. Located at Sta.
6+26.000 @ 907.00 Rt

Dist to Sta 1   611.50
Dist to Sta 2   416.49
Dist to Sta 3   573.53
```

```
626.000195
907.001043
```

These coordinates could then be added to your coordinate file, and a radial inverse program such as "**SPRAY**" (book, "HP42S SURVEYING SOLUTIONS", pages 50 and 51) may be used for layout calculations.

limitations

This program will not pass the E.C. of the curve. The setup point must be opposite a station which is **part of the curve** or on **the back tangent**.

storing the slope stake data

It is pretty easy to print out the data in the field while you slopestake, just point the calculator at the printer each time you reach a catch point. If you want to STORE the shots as you go, you can do so with the little program to the right.*

You can put it in right above "**STA**", and it'll be out of the way. Go to "**STA**" and scroll up to 00 to begin input. When you've finished, the "END" will separate the two programs.

modify "SS"

You'll need two insertions in the program, "**SS**", to make it work. Go to the program, then go to label 12 and enter **program mode** with ☐ R/S. Stroke XEQ and the key that is now corresponding to your new menu label "**STOD**".

Stroke ☐ XEQ . 0 0 0 3 to go to program step 03, and type in 40 STO 40 and exit by stroking EXIT.

```
223 RTN
224▶LBL 12          225 XEQ "STOD"
226 FIX 01
227 CLA             02 XEQ "CL1"
                    03 XEQ "FCL"
04 40
05 STO 40           06 CLST
                    07 SF 01
```

Those insertions should be like the ones shown to the left, for them to work properly. The next thing you need to do is write a program that will recall the data later.

```
00 { 71-Byte Prgm }
01▶LBL "STOD"
02 ISG 40
03 STO ST X
04 CLST
05 FIX 02
06 RCL 31
07 RND
08 ENTER
09 RCL 30
10 RND
11 1E6
12 ÷
13 X<>Y
14 1E3
15 X
16 +
17 STO IND 40
18 ISG 40
19 STO ST X
20 CLST
21 FIX 01
22 RCL 02
23 RND
24 ENTER
25 RCLX 29
26 ABS
27 1E3
28 ÷
29 X<>Y
30 1E3
31 X
32 +
33 STO IND 40
34 ISG 40
35 STO ST X
36 RCL 12
37 STO IND 40
38 END
```

the recall data program

We've called this one "**GETD**", and you should put it into the main menu. You can do this by putting it at the top of "**SS**".

*The calculator uses registers 0-40 + 3 registers for each point stored. In order to store 20 catch points you have to size to at least 0101.

Go to "**SS**", scroll up in the usual manner and type in the new program (below).

```
00 { 204-Byte Prgm }
01▶LBL "GETD"
02 XEQ "FCL"
03 XEQ "CL1"
04 "Number of Shots"
05 ⊢"?↳"
06 PROMPT
07 3
08 X
09 40
10 +
11 1E3
12 ÷
13 41
14 +
15 STO 40
16▶LBL A
17 CLST
18 RCL IND 40
19 ENTER
20 IP
21 1E3
22 ÷
23 STO 37
24 R↓
25 FP
26 X<0?
27 1
28 +
29 1E6
30 X
31 CLA
32 XEQ "STA"
33 ⊢" @ "
34 FIX 02
35 RCL 37
36 X<0?
37 SF 00
38 ABS
39 ARCL ST X
40 FC? 00
41 ⊢" Rt"
42 FS?C 00
43 ⊢" Lt"
44 AVIEW
45 ISG 40
46 STO ST X
47 CLST
48 CLA
49 FIX 01
50 RCL IND 40
51 ENTER
52 FP
53 0
54 X<>Y
55 X<0?
56 SF 00
57 X<0?
58 1
59 +
60 1E3
61 X
62 STO 39
63 R↓
64 FS?C 00
65 R↓
66 IP
67 1E3
68 ÷
69 X>0?
70 "FILL "
71 X<0?
72 "CUT "
73 ABS
74 ARCL ST X
75 ⊢" @ "
76 ARCL 39
77 ISG 40
78 STO ST X
79 AVIEW
80 FIX 02
81 "Elev = "
82 ARCL IND 40
83 AVIEW
84 ADV
85 ISG 40
86 GTO A
87 END
```

Label "**GETD**" can be used to output the station, offset, cut and distance as well as the elevation of the catch point.

It does not output any of the reference data.

When you are finished slope staking for the day, you can turn on the printer and print out the catch points wich were set, by executing "**GETD**". The only prompt is for the number of shots.

Because "GETD" routine will only be executed when you stroke the [illegible] key, the output shots are the ones actually set, not the trial shots.

Proof-read the program carefully and, if you want to try it out, do the examples again.

The output from "**GETD**" for the two catch points set in the examples are shown to the right.

```
11+80.200 @ 44.90 Rt
FILL 6.3 @ 18.9
Elev = 100.60

11+99.800 @ 38.69 Lt
CUT 5.8 @ 8.7
Elev = 111.13
```